
T he strong need for increased 
computational performance in 
science and engineering has led 

to the use of heterogeneous comput-
ing, with GPUs and other accelerators 
acting as coprocessors for arithmetic 
intensive data-parallel workloads.1–4 
OpenCL is a new industry standard 
for task-parallel and data-parallel het-
erogeneous computing on a variety 
of modern CPUs, GPUs, DSPs, and 
other microprocessor designs.5 This 
trend toward heterogeneous comput-
ing and highly parallel architectures 
has created a strong need for software 
development infrastructure in the 
form of parallel programming lan-
guages and subroutine libraries that 
can support heterogeneous comput-
ing on multiple vendors’ hardware 
platforms. To address this, developers 
adapted many existing science and en-
gineering applications to take advan-
tage of multicore CPUs and massively 
parallel GPUs using toolkits such as 
Threading Building Blocks (TBB), 
OpenMP, Compute Unified Device 
Architecture (CUDA),6 and others.7,8 
Existing programming toolkits, how-
ever, were either limited to a single 
microprocessor family or didn’t sup-
port heterogeneous computing.

OpenCL provides easy-to-use  
abstractions and a broad set of  
programming APIs based on past 
successes with CUDA and other 

programming toolkits. OpenCL 
defines core functionality that all 
devices support, as well as optional 
functionality for high-function de-
vices; it also includes an extension 
mechanism that lets vendors expose 
unique hardware features and ex-
perimental programming interfaces 
for application developers’ benefit. 
Although OpenCL can’t mask sig-
nificant differences in hardware archi-
tecture, it does guarantee portability  
and correctness. This makes it much 
easier for developers to start with a 
correctly functioning OpenCL pro-
gram tuned for one architecture 
and produce a correctly function-
ing program optimized for another 
architecture.

The OpenCL  
Programming Model 
In OpenCL, a program is executed on 
a computational device, which can be a 
CPU, GPU, or another accelerator 
(see Figure 1). Devices contain one or 
more compute units (processor cores). 
These units are themselves composed 
of one or more single-instruction 
multiple-data (SIMD) processing ele-
ments (PE) that execute instructions 
in lock-step.

OpenCL Device Management
By providing a common language 
and common programming interfaces 

and hardware abstractions, OpenCL 
lets developers accelerate applications 
with task- or data-parallel computa-
tions in a heterogeneous computing 
environment consisting of the host 
CPU and any attached OpenCL de-
vices. Such devices might or might 
not share memory with the host CPU, 
and typically have a different ma-
chine instruction set. The OpenCL 
programming interfaces therefore as-
sume heterogeneity between the host 
and all attached devices.

OpenCL’s key programming inter-
faces include functions for 

enumerating available target de-•	
vices (CPUs, GPUs, and various 
accelerators);
managing the target devices’ •	
contexts;
managing memory allocations; •	
performing host-device memory •	
transfers; 
compiling the OpenCL programs •	
and kernel functions that the  
devices will execute;
launching kernels on the target •	
devices; 
querying execution progress; and•	
checking for errors. •	

Although developers can compile 
and link OpenCL programs into  
inary objects using off  line com-
pilation methodology, OpenCL  

The OpenCL standard offers a common API for program execution on systems composed of different types  
of computational devices such as multicore CPUs, GPUs, or other accelerators.

encourages runtime compilation  
that lets OpenCL programs run 
natively on the target hardware—
even on platforms unavailable to the 
original software developer. Runtime 
compilation eliminates dependencies 
on instruction sets, letting hardware 
vendors significantly change instruc-
tion sets, drivers, and supporting 
libraries from one hardware genera-
tion to the next.2 Applications that 
use OpenCL’s runtime compilation 
features will automatically take ad-
vantage of the target device’s latest 
hardware and software features with-
out having to recompile the main ap-
plication itself.

Because OpenCL targets a broad 
range of microprocessor designs, it 
must support a multiplicity of pro-
gramming idioms that match the tar-
get architectures. Although OpenCL 
guarantees kernel portability and cor-
rectness across a variety of hardware, 
it doesn’t guarantee that a particular 
kernel will achieve peak performance 
on different architectures; the hard-
ware’s underlying nature might make 
some programming strategies more 
appropriate for particular platforms 
than for others. As an example, a 
GPU-optimized kernel might achieve 
peak memory performance when a 
single work-group’s work-items col-
lectively perform loads and stores, 
whereas a Cell-optimized kernel might 
perform better using a double buffer-
ing strategy combined with calls to  
async_workgroup_copy(). Applica-
tions select the most appropriate ker-
nel for the target devices by querying 
the installed devices’ capabilities and 
hardware attributes at runtime.

OpenCL Device Contexts  
and Memory
OpenCL defines four types of memory 
systems that devices can incorporate:

a large high-latency global memory,•	
a small low-latency read-only con-•	
stant memory,
a shared local memory accessible •	
from multiple PEs within the same 
compute unit, and
a private memory, or device regis-•	
ters, accessible within each PE.

Devices can implement local memory 
using high-latency global memory, 
fast on-chip static RAM, or a shared 
register file. Applications can query 
device attributes to determine prop-
erties of the available compute units 
and memory systems and use them 
accordingly. 

Before an application can compile 
OpenCL programs, allocate device 
memory, or launch kernels, it must 
first create a context associated with 
one or more devices. Because OpenCL 
associates memory allocations with a 
context rather than a specific device, 
developers should exclude devices 
with inadequate memory capacity 
when creating a context, otherwise 
the least-capable device will limit the 
maximum memory allocation. Simi-
larly, they should exclude devices from 
a context if they don’t support features 

that OpenCL programs require to run 
on that context.

 Once a context is created, OpenCL 
programs can be compiled at runtime 
by passing the source code to Open-
CL compilation functions as arrays of 
strings. After an OpenCL program 
is compiled, handles can be obtained 
for the kernel functions contained in 
the program. The kernels can then be 
launched on devices within the Open-
CL context. OpenCL host-device 
memory I/O operations and kernels 
are executed by enqueing them into 
one of the command queues associ-
ated with the target device. 

OpenCL and Modern  
Processor Architectures 
State-of-the-art microprocessors con-
tain several architectural features 
that, historically, have been poorly 
supported or difficult to use in exist-
ing programming languages. This has 
led vendors to create their own pro-
gramming tools, language extensions, 
vector intrinsics, and subroutine li-
braries to close the programmability 
gap created by these hardware fea-
tures. To help clarify the relationship 
between the OpenCL programming 

Figure 1. openCl describes hardware in terms of a hierarchy of devices, compute 
units, and clusters of single instruction multiple data (SIMD) processing elements. 
Before becoming accessible to an application, devices must first be incorporated 
into an openCl context. openCl programs contain one or more kernel functions 
as well as supporting routines that kernels can use.

model and the diversity of poten-
tial target hardware, we compare the  
architectural characteristics of three 
exemplary microprocessor families  
and relate them to key OpenCL  
abstractions and OpenCL program-
ming model features.

Multicore CPUs
Modern CPUs are typically com-
posed of a few high-frequency proces-
sor cores with advanced features such 
as out-of-order execution and branch 
prediction. CPUs are generalists that 
perform well for a wide range of appli-
cations, including latency-sensitive se-
quential workloads and coarse-grained 
task- or data-parallel workloads. 

Because they’re typically used for 
latency sensitive workloads with min-
imal parallelism, CPUs require large 
caches to hide main-memory latency. 
Many CPUs also incorporate small-
scale use of SIMD arithmetic units 
to boost the performance of dense 
arithmetic and multimedia workloads. 
Because conventional programming 
languages like C and Fortran don’t 
directly expose these units, their use 
requires calling vectorized subrou-
tine libraries or proprietary vector 
intrinsic functions, or trial-and-error 
source-level restructuring and auto-
vectorizing compilers. AMD, Apple, 
and IBM provide OpenCL implemen-
tations that target multicore CPUs,  
and support the use of SIMD  

instruction set extensions such as x86 
SSE and Power/VMX (vector multi-
media extensions). The current CPU 
implementations for x86 processors 
often make best use of SSE when 
OpenCL kernels are written with 
explicit use of float4 types. CPU im-
plementations often map all memory 
spaces onto the same hardware cache, 
so a kernel that explicitly uses con-
stant and local memory spaces might 
actually incur more overhead than 
a simple kernel that uses only global 
memory references.

The Cell Processor
The Cell Broadband Engine Architec-
ture (CBEA) is a heterogeneous chip 

architecture consisting of one 64-bit 
Power-compliant PE (PPE), multiple 
Synergistic PEs (SPEs), a memory- 
interface controller, and I/O units, 
connected with an internal high-speed 
bus.9 The PPE is a general-purpose 
processor based on the Power architec-
ture and it’s designed to run conven-
tional OS and control-intensive code to 
coordinate tasks running on SPEs.

The SPE is a SIMD streaming 
processor optimized for massive data 
processing that provides most of 
the Cell systems’ computing power.  
Developers can realize an applica-
tion’s task parallelism using multiple 
SPEs, while achieving data and in-
struction parallelism using the SIMD 
instructions and the SPEs’ dual  

execution pipelines. Each SPE has local 
store, a small, local software-managed 
cache-like fast memory. Applications 
can load data from system memory 
to local store or vice versa using di-
rect memory access (DMA) requests, 
with the best bandwidth achieved 
when both source and destination 
are aligned to 128 bytes. Cell can ex-
ecute data transfer and instructions 
simultaneously, letting application 
developers hide memory latency using 
techniques such as double buffering. 
(We describe the architecture and a 
sample application ported to the Cell 
processor elsewhere.1) 

IBM has recently released an Open-
CL toolkit supporting both the Cell 
and Power processors on the Linux 
platform. The IBM OpenCL imple-
mentation supports the embedded 
profile for the Cell SPUs, and uses 
software techniques to smooth over 
some architectural differences be-
tween the Cell SPUs and conventional 
CPUs. On the Cell processor, global 
memory accesses perform best when 
operands are a multiple of 16 bytes 
(such as an OpenCL float4 type). 
The use of larger vector types such as 
float16 lets the compiler unroll loops, 
further increasing performance. The 
program text and OpenCL local and 
private variables share the 256 Kbyte 
Cell SPU local store, which limits the 
practical work-group size because each 
work-item requires private data stor-
age. The Cell DMA engine performs 
most effectively using double buffer-
ing strategies combined with calls to 
async_workgroup_copy() to load data 
from global memory into local store.

Graphics Processing Units
Contemporary GPUs are composed of 
hundreds of processing units running  
at low to moderate frequency, de-
signed for throughput-oriented latency  

The SIMD clusters execute machine instructions in  

lock-step; branch divergence is handled by executing 

both branch paths and masking off results from inactive 

processing units as necessary.

insensitive workloads. To hide global 
memory latency, GPUs contain small 
or moderate sized on-chip caches and 
extensively use hardware multithread-
ing, executing tens of thousands of 
threads concurrently across the pool of 
processing units. The GPU processing 
units are typically organized in SIMD 
clusters controlled by a single instruc-
tion decoder, with shared access to fast 
on-chip caches and shared memories. 
The SIMD clusters execute machine 
instructions in lock-step; branch di-
vergence is handled by executing both 
branch paths and masking off results 
from inactive processing units as nec-
essary. Using SIMD architecture and 
in-order instruction execution allows 
GPUs to contain many more arithme-
tic units in the same area than tradi-
tional CPUs.2,3

Both AMD and Nvidia have  
released OpenCL implementations 
supporting their respective GPUs. 
These devices require many OpenCL 
work-items and work-groups to fully 
saturate the hardware and hide la-
tency. Nvidia GPUs use a scalar pro-
cessor architecture for individual PEs 
exposed by OpenCL, making them 
highly efficient on most OpenCL data 
types. AMD GPUs use a vector ar-
chitecture, and typically achieve best  
performance when OpenCL work-
items operate on four-element vector 
types (such as float4). In many cases, 
a vectorized OpenCL kernel can per-
form well on x86 CPUs and on AMD 
and Nvidia GPUs, but the resulting 
kernel code might be less readable 
than the scalar equivalent. Differences  
in low-level GPU architecture— 
including variations on what memory 
is cached and what memory access 
patterns create bank conflicts—affect 
kernel optimality. Vendor-provided 
OpenCL literature typically contains 
low-level optimization guidelines.  

In the examples that follow, we re-
frain from detail and focus on general 
OpenCL programming concepts.

An Example OpenCL Kernel
To illustrate the process of moving 
serial code to OpenCL, we’ll use an 
example kernel from the Adaptive 
Poisson-Boltzmann Solver.10 APBS is 
a package for calculating biomolecular 
solvation using the Poisson-Boltzmann 
equation (PBE). The PBE is a popular 
continuum model that describes elec-
trostatic interactions between molecu-
lar solutes. As part of solving the PBE, 
potentials are discretized onto a grid 
that’s larger than the bounding vol-
ume containing the target molecule. 
Under Dirichlet boundary conditions, 
we can solve the potential contribu-
tion of grid points on the grid’s faces 
using the single Debye-Hückel (SDH) 
method or the multiple Debye-Hückel  

(MDH) method. For the MDH 
method, the potential at a grid point i  
located at position ri is given by

with the sum taken over all atoms, 
where α is a prefactor that accounts 
for the system of units and solution 
dielectric values, atom j is located at rj 
and has partial charge qj and size σj, and 
the pairwise distance is rij = |rj − ri|.  
The potential at each grid point is 
effectively the sum of all atomic po-
tential contributions in the molecule. 
The MDH method is inherently 
data-parallel when decomposed over 
grid points because they’re computed 
independently and there are no out-
put conflicts. Figure 2 shows a serial 
MDH algorithm, and Figure 3 shows 
a C implementation. 

Figure 2. Summary of a serial multiple Debye-hückel algorithm. the MDh 
algorithm calculates the total potential at each grid point on a grid’s face, as 
described in equation 1. 

1: for i = 1 to M do {loop over grid points on face}

2:   grid potential ⇐ 0.0 
3:   for j = 1 to N do {loop over all atoms}

4:     grid potential ⇐ grid potential +     

     (potential from atom j)

5:   end for 

6: end for 

7: return grid potential

Figure 3. the innermost loops of a sequential C implementation of the multiple 
Debye-hückel method described in equation 1 and Figure 2.

for ( int igrid=0 ; igrid<ngrid ; igrid++ )  {

   float v=0.0f ;

   for ( int jatom=0 ; jatom<natoms ; jatom++)  {

       dx = gx[ igrid ] − ax[ jatom ] ;

       dy = gy[ igrid ] − ay[ jatom ] ;

       dz = gz[ igrid ] − az[ jatom ] ;

       dist = sqrt ( dx*dx + dy*dy + dz*dz ) ;

       v += pre1 * ( charge [ jatom ] / dist ) *

               exp(−xkappa * ( dist − size [ jatom ] )) /

               ( 1.0 f  +  xkappa  * size [ jatom ] ) ;

    }

    val[ igrid ] = v ;

}

Because the potential at each grid 
point can be calculated independently, 
we can use standard methods (such as 
pthreads or OpenMP) to parallelize 
the earlier example on a CPU. We can 
also perform this type of calculation 
with OpenCL with almost no modifi-
cation. The kernel is simply the inner 
loop over the atoms, as Figure 4 shows. 
In effect, the OpenCL dispatcher be-
comes the outer loop over grid points. 
We set the OpenCL global work-
group size to the total number of 
grid points, and each work-item that 
OpenCL dispatches is responsible for 
calculating a potential using the above 
kernel. Each OpenCL work-item de-
termines its grid point index from its 
global work-item index. 

Figure 4’s code runs approximately 
20 times faster on an Nvidia GeForce 
GTX 285 GPU than the serial code 
on a 2.5 GHz Intel Nehalem CPU. 
For reference, on 16 cores, the parallel 
CPU performance is almost 13 times 
faster. However, the kernel in Figure 4 
doesn’t take advantage of locality of 
concurrent accesses to the atom data. In 
this form, each work-item (grid point)  
is responsible for loading each atom’s 

data (x, y, z, charge, and size) resulting in 
global memory transactions that could 
be avoided by changing the algorithm. 
By taking advantage of each Open-
CL compute unit’s fast on-chip local 
memory, data can be staged in local 
memory and then efficiently broadcast 
to all work-items within the same work-
group. This greatly amplifies the effec-
tive memory bandwidth available to the 
algorithm, improving performance. 
Although the global work-group size 
remains the same, the local work-group 
size increases from 1 to some multiple 
of the hardware SIMD width.

As Figure 5’s example shows, the 
work-group size is limited by the 
amount of data that can be loaded into 
shared memory (typically 16 Kbytes 
on Nvidia GPUs). The on-chip local 
memory is partitioned so that each 
of a work-group’s work-items loads a 
block of the position, charge, and size 
data into local memory at a specific 
offset. Local memory barriers ensure 
that data isn’t overwritten before all 
work-items in a work-group have ac-
cessed it. This coordinated data load-
ing and sharing reduces the number 
of slow global memory accesses.

Another optimization involves us-
ing vector types such as float4 or 
float16, which makes it easier for the 
OpenCL compiler to effectively fill 
very long instruction word (VLIW) 
slots and enables wider memory 
transfer operations. With vector 
types, individual work-items process 
multiple grid points at a time. This 
reduces the global work dimensions 
accordingly, with a corresponding in-
crease in register usage. Calculating 
multiple grid points per work-item  
increases the ratio of arithmetic oper-
ations to memory operations because 
the same atom data is referenced 
multi ple times. On the AMD and 
Nvidia GPUs, the use of vector types 
yields a 20 percent increase in perfor-
mance. On the Cell processor, using 
float16 vectors yields a factor 11 times 
increase in kernel performance.

Figure 5 shows one variation of 
these concepts; many others are 
possible. When compared to the 
original serial CPU code, the ap-
proximate performance increase for 
an IBM Cell blade (using float16) 
is 17 times faster, an AMD Radeon 
5870 GPU is 134 times faster, and 

Figure 4. A simple openCl kernel for the multiple Debye-hückel method. this kernel is similar to the original sequential C 
loops, except that the outer loop over grid points has been replaced by a parallel instantiation of independent grid points as 
openCl work-items, and the igrid index is determined by the openCl work-item index.

  __kernel void mdh (__global float *ax , __global float *ay ,

         __global float *az , __global float *charge ,

         __global float *size , __global float *gx ,

         __global float *gy , __global float *gz ,

         __global float *val ,

         float pre1 , float xkappa ,  cl _int natoms)  {

  int igrid = get_global_id ( 0 ) ;

  float v = 0.0f ;

  for (int jatom = 0 ; jatom < natoms ; jatom++)  {

    float dx = gx [ igrid ] − ax [ jatom ] ;

    float dy = gy [ igrid ] − ay [ jatom ] ;

    float dz = gz [ igrid ] − az [ jatom ] ;

    float dist = sqrt( dx*dx + dy*dy + dz*dz) ;

    v += pre1 * ( charge [ jatom ] / dist ) *

         exp( −xkappa * ( dist − size [ jatom ] ) )   /

         (1.0 f  +  xkappa * size [ jatom ] ) ;

  }

  val [ igrid ] = v ;

}

an Nvidia GeForce GTX 285 GPU 
is 141 times faster. With further 
platform-specific tuning, each of 
these platforms could undoubtedly 
achieve even higher performance. A 
more important outcome still is that 
the numerical result is exact—within  
the floating point rounding mode 

and summation order—to the CPU 
methods (both serial and parallel). 

Our initial experiences in adapt-
ing molecular modeling appli-

cations such as APBS and VMD11 
to OpenCL 1.0 have been gener-
ally positive. In the coming year, we  

expect OpenCL to incorporate new 
features and promote previously 
optional features to core features  
of OpenCL 1.1 and later versions. 
Recent GPUs allow direct access to  
host memory over peripheral compo-
nent interconnect express (PCI-E), 
and some enable global GPU memory 

Figure 5. the optimized openCl kernel for the multiple Debye-hückel method. this kernel is similar to the simple openCl 
kernel in Figure 4, but each work-group collectively loads and processes blocks of atom data in fast on-chip local memory. 
openCl barrier instructions enforce coordination between the loading and processing phases to maintain local memory 
consistency. the inner loop uses vector types to process multiple grid points per work-item, and atom data is processed 
entirely from on-chip local memory, greatly increasing both arithmetic intensity and effective memory bandwidth.

typedef float4 floatvec;

__kernel void mdh(__global float *ax, __global float *ay, 

                  __global float *az, __global float *charge, 

                  __global float *size, __global floatvec *gx, 

                  __global floatvec *gy, __global floatvec *gz,

                  __global floatvec *val, __local float *shared,

                  float pre1, float xkappa, cl_int natoms) {

    int igrid = get_global_id(0);

    int lsize = get_local_size(0);

    int lid = get_local_id(0);

    floatvec lgx = gx[igrid];

    floatvec lgy = gy[igrid];

    floatvec lgz = gz[igrid];

    floatvec v = 0.0f;

    for (int jatom = 0; jatom < natoms; jatom+=lsize) {

        if ((jatom+lsize) > natoms) lsize = natoms - jatom;

 

        if ((jatom + lid) < natoms) {

            shared[lid          ] = ax[jatom + lid];

            shared[lid +   lsize] = ay[jatom + lid];

            shared[lid + 2*lsize] = az[jatom + lid];

            shared[lid + 3*lsize] = charge[jatom + lid];

            shared[lid + 4*lsize] = size[jatom + lid];

        }

        barrier(CLK_LOCAL_MEM_FENCE);

 

        for (int i=0; i<lsize; i++) {

            floatvec dx = lgx - shared[i          ];

            floatvec dy = lgy - shared[i +   lsize];

            floatvec dz = lgz - shared[i + 2*lsize];

            floatvec dist = sqrt(dx*dx + dy*dy + dz*dz);

            v += pre1 * (charge[jatom] / dist)  *

                 exp(-xkappa * (dist - size[jatom])) /

                 (1.0f + xkappa * size[jatom]);

        }

        barrier(CLK_LOCAL_MEM_FENCE);

    }

    val[igrid] = v;

}

mapping into the host address space, 
allowing fast zero-copy access to data 
that’s read or written only once during 
kernel execution. Although zero-copy 
APIs exist in CUDA, OpenCL cur-
rently doesn’t include this capability. 
As OpenCL matures, we hope to see 
increased support for thread-safety, 
increased OpenGL interoperability, 
and extension with advanced features 
found in APIs such as CUDA. 

OpenCL provides correctness guar-
antees such that code written and opti-
mized for one device will run correctly 
on any other device, although not nec-
essarily with peak performance. The 
only exception to this is when optional 
OpenCL features are used that aren’t 
available on the target device. As future 
hardware platforms that operate on 
wider vectors—such as Intel’s advanced 
vector extensions (AVX)—arrive, we’re 
eager to see OpenCL implementa-
tions incorporate a greater degree of 
autovectorization, enabling efficient 
kernels to be written with less vector 
width specificity, and improving the 
performance portability of OpenCL 
kernels across the board. Our experi-
ences show that OpenCL holds great 
promise as a standard low-level parallel 
programming interface for heteroge-
neous computing devices. 

